

# Verilog generate block

A `generate` block allows to multiply module instances or perform conditional instantiation of any module. It provides the ability for the design to be built based on Verilog parameters. These statements are particularly convenient when the same operation or module instance needs to be repeated multiple times or if certain code has to be conditionally included based on given Verilog parameters.

A `generate` block cannot contain port, parameter, `specparam` declarations or `specify` blocks. However, other module items and other generate blocks are allowed. All generate instantiations are coded within a `module` and between the keywords `generate` and `endgenerate`.

Generated instantiations can have either modules, continuous assignments, `always` or `initial` blocks and user defined primitives. There are two types of generate constructs - loops and conditionals.

- Generate for loop
- Generate if else
- Generate case

## Generate for loop

A half adder will be instantiated `N` times in another top level design module called `my_design` using a `generate` for loop construct. The loop variable has to be declared using the keyword `genvar` which tells the tool that this variable is to be specifically used during elaboration of the generate block.

```

1 // Design for a half-adder
2 module ha ( input  a, b,
3             output sum, cout);
4
5     assign sum  = a ^ b;
6     assign cout = a & b;
7 endmodule
8
9 // A top level design that contains N instances of half adder
10 module my_design
11     #(parameter N=4)
12     (   input [N-1:0] a, b,
13         output [N-1:0] sum, cout);
14
15     // Declare a temporary loop variable to be used during
16     // generation and won't be available during simulation
17     genvar i;
18
19     // Generate for loop to instantiate N times
20     generate
21         for (i = 0; i < N; i = i + 1) begin
22             ha u0 (a[i], b[i], sum[i], cout[i]);
23         end
24     endgenerate
25 endmodule

```

## Testbench

The testbench parameter is used to control the number of half adder instances in the design. When

**N** is 2, **my\_design** will have two instances of half adder.

```

1 module tb;
2     parameter N = 2;
3     reg  [N-1:0] a, b;
4     wire [N-1:0] sum, cout;
5
6     // Instantiate top level design with N=2 so that it will have 2
7     // separate instances of half adders and both are given two separate
8     // inputs
9     my_design #(.N(N)) md( .a(a), .b(b), .sum(sum), .cout(cout));
10
11     initial begin
12         a <= 0;
13         b <= 0;
14
15         $monitor ("a=0x%0h b=0x%0h sum=0x%0h cout=0x%0h", a, b, sum, cout);
16
17         #10 a <= 'h2;
18             b <= 'h3;
19         #20 b <= 'h4;
20         #10 a <= 'h5;
21     end
22 endmodule

```

`a[0]` and `b[0]` gives the output `sum[0]` and `cout[0]` while `a[1]` and `b[1]` gives the output `sum[1]` and `cout[1]`.


## Simulation Log

```

ncsim> run
a=0x0 b=0x0 sum=0x0 cout=0x0
a=0x2 b=0x3 sum=0x1 cout=0x2
a=0x2 b=0x0 sum=0x2 cout=0x0
a=0x1 b=0x0 sum=0x1 cout=0x0
ncsim: *W,RNQUIE: Simulation is complete.
ncsim> exit

```

See that elaborated RTL does indeed have two half adder instances generated by the `generate` block.



([https://www.chipverify.com/images/verilog/schematic/generate\\_block\\_for\\_loop\\_ha\\_schematic.png](https://www.chipverify.com/images/verilog/schematic/generate_block_for_loop_ha_schematic.png)).

## Generate if

Shown below is an example using an `if else` inside a `generate` construct to select between two different multiplexer implementations. The first design uses an `assign` statement to implement a mux while the second design uses a `case` statement. A parameter called `USE_CASE` is defined in the top level design module to select between the two choices.

```

1 // Design #1: Multiplexer design uses an "assign" statement to assign
2 // out signal
3 module mux_assign ( input a, b, sel,
4                     output out);
5     assign out = sel ? a : b;
6
7     // The initial display statement is used so that
8     // we know which design got instantiated from simulation
9     // logs
10    initial
11        $display ("mux_assign is instantiated");
12 endmodule
13
14 // Design #2: Multiplexer design uses a "case" statement to drive
15 // out signal
16 module mux_case (input a, b, sel,
17                 output reg out);
18     always @ (a or b or sel) begin
19         case (sel)
20             0 : out = a;
21             1 : out = b;
22         endcase
23     end
24
25     // The initial display statement is used so that
26     // we know which design got instantiated from simulation
27     // logs
28     initial
29         $display ("mux_case is instantiated");
30 endmodule
31
32 // Top Level Design: Use a parameter to choose either one
33 module my_design ( input a, b, sel,
34                   output out);
35     parameter USE_CASE = 0;
36
37     // Use a "generate" block to instantiate either mux_case
38     // or mux_assign using an if else construct with generate
39     generate
40         if (USE_CASE)
41             mux_case mc (.a(a), .b(b), .sel(sel), .out(out));
42         else
43             mux_assign ma (.a(a), .b(b), .sel(sel), .out(out));
44     endgenerate
45
46 endmodule

```

## Testbench

Testbench instantiates the top level module `my_design` and sets the parameter `USE_CASE` to 1 so that it instantiates the design using `case` statement.

```
1 module tb;
2     // Declare testbench variables
3     reg a, b, sel;
4     wire out;
5     integer i;
6
7     // Instantiate top level design and set USE_CASE parameter to 1 so that
8     // the design using case statement is instantiated
9     my_design #(.USE_CASE(1)) u0 ( .a(a), .b(b), .sel(sel), .out(out));
10
11     initial begin
12         // Initialize testbench variables
13         a <= 0;
14         b <= 0;
15         sel <= 0;
16
17         // Assign random values to DUT inputs with some delay
18         for (i = 0; i < 5; i = i + 1) begin
19             #10 a <= $random;
20             b <= $random;
21             sel <= $random;
22             $display ("i=%0d a=0x%0h b=0x%0h sel=0x%0h out=0x%0h", i, a, b, sel, out);
23         end
24     end
25 endmodule
```

When the parameter `USE_CASE` is 1, it can be seen from the simulation log that the multiplexer design using `case` statement is instantiated. And when `USE_CASE` is zero, the multiplexer design using `assign` statement is instantiated. This is visible from the display statement that gets printed in the simulation log.

## Simulation Log

```
// When USE_CASE = 1
ncsim> run
mux_case is instantiated
i=0 a=0x0 b=0x0 sel=0x0 out=0x0
i=1 a=0x0 b=0x1 sel=0x1 out=0x1
i=2 a=0x1 b=0x1 sel=0x1 out=0x1
i=3 a=0x1 b=0x0 sel=0x1 out=0x0
i=4 a=0x1 b=0x0 sel=0x1 out=0x0
ncsim: *W,RNQUIE: Simulation is complete.
```

```
// When USE_CASE = 0
ncsim> run
mux_assign is instantiated
i=0 a=0x0 b=0x0 sel=0x0 out=0x0
i=1 a=0x0 b=0x1 sel=0x1 out=0x0
i=2 a=0x1 b=0x1 sel=0x1 out=0x1
i=3 a=0x1 b=0x0 sel=0x1 out=0x1
i=4 a=0x1 b=0x0 sel=0x1 out=0x1
ncsim: *W,RNQUIE: Simulation is complete.
```

## Generate Case

A generate case allows modules, initial and always blocks to be instantiated in another module based on a `case` expression to select one of the many choices.

```

1 // Design #1: Half adder
2 module ha (input a, b,
3           output reg sum, cout);
4     always @ (a or b)
5     {cout, sum} = a + b;
6
7     initial
8     $display ("Half adder instantiation");
9 endmodule
10
11 // Design #2: Full adder
12 module fa (input a, b, cin,
13           output reg sum, cout);
14     always @ (a or b or cin)
15     {cout, sum} = a + b + cin;
16
17     initial
18     $display ("Full adder instantiation");
19 endmodule
20
21 // Top level design: Choose between half adder and full adder
22 module my_adder (input a, b, cin,
23                 output sum, cout);
24     parameter ADDER_TYPE = 1;
25
26     generate
27     case(ADDER_TYPE)
28     0 : ha u0 (.a(a), .b(b), .sum(sum), .cout(cout));
29     1 : fa u1 (.a(a), .b(b), .cin(cin), .sum(sum), .cout(cout));
30     endcase
31     endgenerate
32 endmodule

```

## Testbench



```

1 module tb;
2     reg a, b, cin;
3     wire sum, cout;
4
5     my_adder #(.ADDER_TYPE(0)) u0 (.a(a), .b(b), .cin(cin), .sum(sum), .cout(cout));
6
7     initial begin
8         a <= 0;
9         b <= 0;
10        cin <= 0;
11
12        $monitor("a=0x%0h b=0x%0h cin=0x%0h cout=0%0h sum=0x%0h",
13                a, b, cin, cout, sum);
14
15        for (int i = 0; i < 5; i = i + 1) begin
16            #10 a <= $random;
17            b <= $random;
18            cin <= $random;
19        end
20    end
21 endmodule

```

Note that because a half adder is instantiated, `cin` does not have any effect on the outputs `sum` and `cout`.

## Simulation Log

```

ncsim> run
Half adder instantiation
a=0x0 b=0x0 cin=0x0 cout=00 sum=0x0
a=0x0 b=0x1 cin=0x1 cout=00 sum=0x1
a=0x1 b=0x1 cin=0x1 cout=01 sum=0x0
a=0x1 b=0x0 cin=0x1 cout=00 sum=0x1
ncsim: *W,RNQUIE: Simulation is complete.

```